Извините, но безымянные работы не проверяются. 
Я все же заглянула дальше… Вы очень любезны. Вот что выдает Ваша программа: 

В связи с этим, для получения положительной оценки Вам даётся дополнительное задание. Следует для того же задания рассмотреть ещё и леволинейную грамматику. Т.е. персонально Вам как лицу, склонному к плагиату, снимается предоставленная в условии льгота «При написании программы разработчику разрешается выбрать один из двух типов регулярной грамматики (ЛЛ или ПЛ) и следует информировать об этом пользователя.»

                                    1. ПОСТАНОВКА ЗАДАЧИ

Вариант 4 

         Написать программу для автоматического построения детерминированного конечного автомата (ДКА), эквивалентного заданной регулярной грамматике. 

         Вход программы: терминальный и нетерминальный алфавиты грамматики, целевой символ, правила грамматики, цепочки для распознавания. 

         Выход: построенный ДКА (все 5 элементов), результат проверки цепочек.

        Описание:

         Язык задан регулярной грамматикой, причём она может быть не автоматного вида. При написании программы разработчику разрешается выбрать один из двух типов регулярной грамматики (ЛЛ или ПЛ) и следует информировать об этом пользователя. Терминальный алфавит грамматики может включать в себя любые символы, в нетерминальном алфавите могут использоваться заглавные буквы латинского алфавита или (на усмотрение разработчика) слова. Правила задаваемой грамматики должны соответствовать выбранному типу. Для того чтобы в исходной грамматике можно было использовать пустое правило, необходимо предусмотреть поле ввода для символа, которым пользователь может обозначить пустую цепочку.

Программа должна:

1. по заданной регулярной грамматике строить эквивалентный ДКА, распознающий этот же язык, в том виде, как он рассматривался в теории, раздел 2.2.2;

2. с помощью построенного ДКА проверять вводимые пользователем цепочки на их принадлежность этому языку.

ДКА должен распознавать язык, задаваемый исходной грамматикой, т.е. являться эквивалентной конструкцией. Функция переходов ДКА может изображаться в виде таблицы или графа, вариант вида её представления выбирается разработчиком. Для удобства построения автомата рекомендуется предварительно привести заданную грамматику к автоматному виду (в соответствии с лекционным разделом 2.2.1).

При выборе такого способа построения ДКА, когда сначала по заданной грамматике строится эквивалентный НКА, а затем он приводится к детерминированному виду, промежуточный результат в виде НКА необходимо также отображать на экране по просьбе пользователя.

После построения ДКА пользователь может вводить произвольные цепочки для проверки их на принадлежность исходному языку. Разбор цепочек автоматом следует поэтапно отображать на экране в виде последовательной смены конфигураций в соответствии с лабораторной работой №2.

2. ОПИСАНИЕ АЛГОРИТМА РЕШЕНИЯ ЗАДАЧИ
Решение задачи построения ДКА будем выполнять только для  праволинейных грамматик, в правилах которых нетерминальные символы могут следовать только после терминальных символов. В программе введены также ограничения на размер нетерминального алфавита грамматики в 26 символов и на размер терминального алфавита грамматики в 8 символов. Для задания пустого символа  предусмотрено специальное поле, которое может быть заполнено пользователем. По умолчанию, в программе в качестве пустого символа используется «#».  Символ, который установлен в качестве пустого, не может быть указан в терминальном и нетерминальном алфавите грамматики, будет выдано сообщение об ошибке.
Правила грамматики задаются в специальном поле формы в следующем формате:

-указывается нетерминальный символ, стоящий в левой части правила;

-символ «=»;

-записывается строка терминальных и нетерминальных символов, расположенных в правой части соответствующего правила;

-правила, относящиеся к одному и тому же нетерминальному символу, отделяются друг от друга символом «|»
-конец правил, относящихся к одному и тому же нетерминальному символу, задается символом «;»
-за символом «;» следуют правила для других нетерминальных символов, записываемых в таком же формате, как и указано выше;

-символ «;» за последним правилом грамматики указывать не обязательно;

-посторонние пробелы в записи правил не допускаются. 

Построение ДКА, эквивалентного заданной праволинейной регулярной грамматике, разбивается на три этапа. На первом этапе строится НКА, который затем проверяется на детерминированность. Если проверка оказывается успешной, то следующие два этапа выполнять не надо. На втором этапе происходит преобразование НКА в эквивалентный ДКА. Третий этап заключается в удалении из построенного ДКА недостижимых состояний. После завершения третьего этапа все состояния ДКА, состоящие более чем из одного символа, заменяются на заглавные буквы латинского алфавита, которые не использовались в данном ДКА в качестве имен состояний. Это обеспечивает большую наглядность таблицы переходов ДКА.
Автоматная грамматика не строится, т.к. собственно правила автоматной грамматики не формируются. Ниже, в пункте 4 этапа 1 содержится описание, как происходит построение автомата, когда правило исходной грамматики не является автоматным.

Рассмотрим далее все три этапа.

Этап 1 – построение НКА, эквивалентного исходной грамматике.
Состояниями автомата становятся все символы нетерминального алфавита грамматики. Все символы терминального алфавита грамматики становятся множеством терминальных символов автомата. Состояние автомата, соответствующее начальному символу грамматики становится начальным состоянием автомата. Затем последовательно происходит разбор каждого отдельного правила грамматики. При этом программа контролирует отсутствие ошибок в анализируемых правилах. Далее при описании алгоритма разбора правил для простоты будем считать, что в них отсутствуют недопустимые конструкции. 
Программа анализирует цепочку символов правой части правила на глубину двух символов. При этом возможны следующие ситуации:

1) Правая часть правила – пустая цепочка. Текущее состояние автомата объявляется конечным. Разбор данного правила окончен.

2) Правая часть правила состоит из одного терминального символа. Проверяется, добавлялось ли в автомат конечное состояние, из которого отсутствуют переходы в другие состояния. Если такого состояния нет, то к НКА добавляется новое конечное состояние, из которого нет переходов в другие состояния автомата. Строится переход из текущего состояния в это конечное состояние. Разбор данного правила окончен.

3) Правая часть правила состоит из терминального символа, за которым следует нетерминальный. Строится переход из текущего состояния в состояние, соответствующее нетерминальному символу. Разбор данного правила окончен.

4) Первые два символа правой части правила являются терминальными символами. К НКА добавляется новое состояние, в которое строится переход из текущего состояния по первому терминальному символу. Добавленное состояние становится текущим состоянием, т.е. становясь нетерминалом в левой части правила. Из цепочки, представляющей собой правую часть правила, удаляется первый терминальный символ.  Разбор оставшейся части данного правила продолжается по правилам 1-4.
После обработки по описанным правилам всех правил исходной грамматики будет построен искомый НКА.
Проверка автомата на детерминированность выполняется путем проверки наличия в автомате хотя бы одного состояния, из которого есть переходы в несколько состояний по одинаковому терминальному символу.

Этап 2 – преобразование НКА в эквивалентный ДКА.

Преобразование НКА осуществляется в следующем порядке:

1) Просматривается таблица переходов автомата для поиска элементов, в которых записаны переходы в несколько состояний по одному терминальному символу. Если таких элементов не найдено, то этап 2 завершен.
2) К автомату добавляется новое состояние, для имени которого используется конкатенация символов состояний, находящихся в элементе таблицы переходов. В элемент таблицы переходов вместо нескольких переходов записывается новое имя состояния. Для нового состояния строятся переходы по всем терминальным символам как совокупность переходов из всех состояний, образовавших новое состояние.

3) Переход к пункту 1.

По окончании этапа 2 будет построен ДКА, эквивалентный НКА. Возможно, что в нем будут присутствовать недостижимые состояния.
Этап 3 – удаление недостижимых символов

Для удаления недостижимых символов используется алгоритм, который для каждого состояния автомата проверяет его присутствие, хотя бы в одном элементе таблицы переходов. Если для некоторого состояния определяется, что нет его вхождений в таблицу переходов, то данное состояние является недостижимым, и оно удаляется из автомата. Возможно такая методика неверная, и состояние может неоднократно попадать в таблицу переходов, но предложенный метод удаляет все лишние состояния.
Действие, описанное выше, повторяется до тех пор, пока проверка всех оставшихся состояний автомата не выявит ни одного недостижимого состояния.
Рассмотрим пример построения ДКА по описанному алгоритму.
Пусть дана праволинейная грамматика: G({0,1},{S},P,S), 
где Р: S=0S|1S|11. Данная грамматика задает генерацию цепочек, состоящую из символов «0» и «1», заканчивающихся цепочкой «11».
          Этап 1.

Множество состояний автомата = {S}, множество терминальных символов = {0,1}, начальное состояние автомата S.
Разбираем правило S=0S. Оно соответствует пункту 3 алгоритма этапа 1. В таблице переходов строится переход из состояния S в S по символу «0». На этом разбор данного правила заканчивается.
Разбираем правило S=1S. Оно соответствует пункту 3 алгоритма этапа 1. В таблице переходов строится переход из состояния S в S по символу «1». На этом разбор данного правила заканчивается.

Разбираем правило S=11. Оно соответствует пункту 4 алгоритма этапа 1. К автомату добавляется новое состояние A (первый заглавный символ латинского алфавита, который на данный момент не встречается во множестве состояний). В таблице переходов строится переход из состояния S в A по символу «1». Правило принимает вид A=1.
Разбираем правило A=1. Оно соответствует пункту 2 алгоритма этапа 1. Поскольку к автомату не добавлялось конечное состояние, из которого нет переходов в другие состояния, то к автомату добавляется новое состояние B (первый заглавный символ латинского алфавита, который на данный момент не встречается во множестве состояний). Состояние B добавляется во множество конечных состояний. В таблице переходов строится переход из состояния A в B по символу «1». Правило принимает вид A=1. На этом разбор данного правила заканчивается.
Все правила грамматики разобраны, этап 1 завершен, НКА построен:
M({S,A,B},{0,1},d,S,{B})
функция переходов d:
	
	0
	1

	S
	{S}
	{S,A}

	A
	
	{B}

	B
	
	


Проверка автомата на детерминированность дает отрицательный результат, т.к. в таблице переходов присутствует элемент с двумя переходами {S,A} по терминальному символу «1».
          Этап 2.

Согласно пункту 1 просматривая таблицу переходов, находим элемент с двумя переходами {S,A} по терминальному символу «1».
Согласно пункту 2 добавляем к автомату новое состояние SA, заносим его имя в элемент таблицы с двумя переходами,  в таблице переходов строим переходы для всех терминальных символов.

	
	0
	1

	S
	{S}
	{SA}

	A
	
	{B}

	B
	
	

	SA
	{S}
	{SA,B}


Согласно пункту 1 просматривая таблицу переходов, находим элемент с двумя переходами {SA,B} по терминальному символу «1».
Согласно пункту 2 добавляем к автомату новое состояние SAB, заносим его имя в элемент таблицы с двумя переходами,  в таблице переходов строим переходы для всех терминальных символов. Поскольку состояние B являлось конечным, то состояние SAB заносится во множество конечных символов.
	
	0
	1

	S
	{S}
	{SA}

	A
	
	{B}

	B
	
	

	SA
	{S}
	{SAB}

	SAB
	{S}
	{SAB}


Согласно пункту 1 просматривая таблицу переходов не находим ни одного элемента с двумя или более переходами по одному терминальному символу. ДКА построен.

          Этап 3.

Просматривая таблицу переходов для всех состояний автомата, не находим ни одного перехода в состояние A. Следовательно, оно является недостижимым, удаляем его из автомата.

Просматривая новую таблицу переходов для всех состояний автомата, не находим ни одного перехода в состояние B. Следовательно, оно является недостижимым, удаляем его из автомата.
Просматривая новую таблицу переходов для всех состояний автомата, не находим состояний, в которые нет переходов. Удаление недостижимых состояний завершено.

После завершения третьего этапа все состояния ДКА, состоящие более чем из одного символа, заменяются на заглавные буквы латинского алфавита, которые не использовались в данном ДКА в качестве имен состояний.

	
	0
	1

	S
	{S}
	{SA}

	SA
	{S}
	{SAB}

	SAB
	{S}
	{SAB}


Состояние SA состоит более чем из одного символа, первым неиспользованным заглавным символом латинского алфавита является A. Заменяем SA на A.

Состояние SAB состоит более чем из одного символа, первым неиспользованным заглавным символом латинского алфавита является B. Заменяем SAB на B.
Переименование состояний завершено, ДКА построен:
M({S,A,B},{0,1},d,S,{B})
функция переходов d:
	
	0
	1

	S
	{S}
	{A}

	A
	{S}
	{B}

	B
	{S}
	{B}



После окончания построения автомата на экран будут выведены все пять элементов ДКА: множество состояний, множество символов алфавита, таблица переходов, начальное состояние, множество конечных состояний. Одновременно активизируется кнопка «Проверить цепочку».
          Чтение цепочки автоматом отображается на экране в виде последовательной смены конфигураций в соответствии с лабораторной работой №2. При возникновении ошибок при чтении входной цепочки программа прекращает работу и выводит на экран сообщение об ошибке.

3. ОПИСАНИЕ ОСНОВНЫХ ПЕРЕМЕННЫХ И БЛОКОВ
Глобальные переменные:
-правила переходов автомата:-  P: Array [1..26,1..8] of String;

-состояния автомата: VN : Array [1..26] of String;
-массив признаков принадлежности состояния множеству конечных состояний: Final: Array [1..26] of Boolean;

-терминальный алфавит грамматики: - SVT: String;
-начальный символ грамматики S0, String;
- пустой символ - SEmpty: String;
- правила грамматики - SP: String;
-нетерминальный алфавит грамматики - SVN: String;
-терминальный алфавит грамматики - SVT: String;

-индекс конечного состояния автомата без переходов из него - IdxPFinal: Integer;

Процедуры и функции:

-функция поверки автомата на детерминированность:

    function CheckDFA: Boolean;

-функция удаления недостижимого состояния автомата:

    function DeleteUnavailable: Boolean;

-функция преобразования НКА в ДКА:

    procedure NFAToDFA;

-процедура вывода автомата на форму:

    procedure OutAutomat;
-функция разбора правила грамматики и преобразования в переходы автомата:
    function ParseRule(IdxLVN: Integer; var I: Integer; L: Integer): Boolean;

-функция преобразования грамматики в НКА:

    function RGToNFA: Boolean;
4. ТЕКСТ ПРОГРАММЫ
Unit1.pas
unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,

  Dialogs, Math, StdCtrls, Grids, Menus, ExtCtrls;

type

  TForm1 = class(TForm)

    Button1: TButton;

    Label1: TLabel;

    Label2: TLabel;

    Label3: TLabel;

    Label4: TLabel;

    Label5: TLabel;

    Edit1: TEdit;

    Edit2: TEdit;

    Edit3: TEdit;

    Edit5: TEdit;

    ListBox1: TListBox;

    StringGrid1: TStringGrid;

    MainMenu1: TMainMenu;

    N1: TMenuItem;

    N2: TMenuItem;

    N3: TMenuItem;

    Edit4: TEdit;

    Edit6: TEdit;

    Edit7: TEdit;

    Label6: TLabel;

    N4: TMenuItem;

    N5: TMenuItem;

    Label7: TLabel;

    Edit8: TEdit;

    Label8: TLabel;

    Label9: TLabel;

    Label10: TLabel;

    Edit9: TEdit;

    Label11: TLabel;

    procedure Button1Click(Sender: TObject);

    procedure OnN3(Sender: TObject);

    procedure OnN4(Sender: TObject);

    procedure OnN2(Sender: TObject);

    procedure OnN5(Sender: TObject);

  private

    function AddVN(P, S: String): String;

    function CheckDFA: Boolean;

    procedure CheckFinal(IdxVN: Integer);

    function CheckState(S: String): Boolean;

    function DeleteUnavailable: Boolean;

    function GetNewVNSymbol: Char;

    procedure NFAToDFA;

    procedure OutAutomat;

    function ParseRule(IdxLVN: Integer; var I: Integer; L: Integer): Boolean;

    procedure ReadGrammar;

    procedure RenameVN;

    function RGToNFA: Boolean;

    function SortString(S: String): String;

  public

    { Public declarations }

  end;

var

 Form1: TForm1;

 P: Array [1..26,1..8] of String; //таблица переходов

 VN: Array [1..26] of String; //состояния автомата

 Final: Array [1..26] of Boolean; //признак, что i-е состояние автомата конечное

 S0: String; //начальный символ грамматики

 SP: String; //правила грамматики

 SEmpty: String; //пустой символ

 SVN, SVT: String; //нетерминальный алфавит, терминальный алфавит грамматики

 NVN: Integer; //число нетерминальных символов

 IdxPFinal: Integer;//индекс конечного состояния автомата без переходов из него

implementation
{$R *.dfm}

function TForm1.AddVN(P, S: String): String; //добавление нетерминала в алфавит нетерминалов

var I: Integer;

begin

 for I:=1 to Length(S) do

  if Pos(S[I],P)=0 then P:=P+S[I];

 AddVN:=SortString(P);

end;

function TForm1.CheckDFA: Boolean; //функция проверки, является ли автомат детерминированным

var I, J: Integer;

begin

 for I:=1 to NVN do

  for J:=1 to Length(SVT) do

   if Length(P[I,J])>1 then begin

    CheckDFA:=FALSE;

    exit;

   end;

 CheckDFA:=TRUE;

end;

//процедура проверки является ли автомата конечным

procedure TForm1.CheckFinal(IdxVN: Integer);

var I: Integer;

begin

 Final[IdxVN]:=TRUE;

 for I:=1 to Length(SVT) do

  if (P[IdxVN,I]<>'') then begin

   Final[IdxVN]:=FALSE;

   exit;

  end;

end;

function TForm1.CheckState(S: String): Boolean; //функция проверки наличия состояния S во множестве состояний автомата

var I: Integer;

begin

 for I:=1 to NVN do

  if S=VN[I] then begin

   CheckState:=TRUE;

   exit;

  end;

 CheckState:=FALSE;

end;

function TForm1.DeleteUnavailable: Boolean; //функция удаления недостежимых состояний из ДКА

var I, J, K: Integer; FFind: Boolean;

begin

 for I:=1 to NVN do begin

  if VN[I]=S0 then continue;

  FFind:=FALSE;

  for J:=1 to NVN do

   for K:=1 to Length(SVT) do

    if VN[I]=P[J,K] then FFind:=TRUE;

  if Not FFind then begin

   Dec(NVN);

   for J:=I to NVN do begin

    Final[J]:=Final[J+1];

    VN[J]:=VN[J+1];

    for K:=1 to Length(SVT) do P[J,K]:=P[J+1,K];

   end;

   DeleteUnavailable:=TRUE;

   exit;

  end;

 end;

 DeleteUnavailable:=FALSE;

end;

function TForm1.GetNewVNSymbol: Char;//функция формирования имени очередного нетерминала

var I, J: Integer; Ch: Char; FExist: Boolean;

begin

 for I:=0 to 25 do begin

  Ch:=Chr(Ord('A')+I);

  FExist:=FALSE;

  for J:=1 to NVN do

   if SVN[J]=Ch then FExist:=TRUE;

  if Not FExist then begin

   GetNewVNSymbol:=Ch;

   exit;

  end;

 end;

 ShowMessage('Нет места для порождения нового нетерминального символа');

 GetNewVNSymbol:=#0;

end;

procedure TForm1.NFAToDFA;//процедура преобразования НКА в ДКА

var I, J, K, L: Integer; FDo: Boolean;

begin

 FDo:=TRUE;

 while (FDo) do begin

  FDo:=FALSE;

  I:=1;

  while ((Not FDo) And (I<=NVN)) do begin

   J:=1;

   while ((Not FDo) And (J<=Length(SVT))) do

    if ((P[I,J]<>'') And (Not CheckState(P[I,J]))) then FDo:=TRUE

                                                   else Inc(J);

   if Not FDo then Inc(I);

  end;

  if FDo then begin

   Inc(NVN);

   VN[NVN]:=P[I,J];

   for I:=1 to Length(VN[NVN]) do begin

    L:=Pos(VN[NVN][I],SVN);

    Final[NVN]:=Final[NVN] Or Final[L];

    for K:=1 to Length(SVT) do

     if P[L,K]<>'' then P[NVN,K]:=AddVN(P[NVN,K],P[L,K]);

   end;

  end;

 end;

end;

procedure TForm1.OutAutomat; //процедура вывода автомата на форму

var I, J: Integer; S, SF: String;

begin

 SF:='';

 StringGrid1.ColCount:=Length(SVT)+1;

 StringGrid1.RowCount:=NVN+1;

 for J:=1 to Length(SVT) do StringGrid1.Cells[J,0]:=SVT[J];

 for I:=1 to NVN do begin

  StringGrid1.Cells[0,I]:=VN[I];

  for J:=1 to Length(SVT) do StringGrid1.Cells[J,I]:=P[I,J];

  if Final[I] then SF:=SF+VN[I];

 end;

 S:='';

 for I:=1 to NVN do S:=S+VN[I];

 Edit4.Text:=S;

 Edit6.Text:=S0;

 Edit7.Text:=SF;

end;

//функция разбора очередного правила грамматики

function TForm1.ParseRule(IdxLVN: Integer; var I: Integer; L: Integer): Boolean;

var IdxVN, IdxVT, J: Integer; Ch: Char; FVN: Boolean;

begin

 FVN:=FALSE;

 IdxVN:=IdxLVN;

 IdxVT:=0;

 while ((I<=L) And (SP[I]<>'|') And (SP[I]<>';')) do begin

  J:=Pos(SP[I],SVT);

  if J=0 then begin

   if IdxVT=0 then begin

    ShowMessage('В правиле пропущен терминальный символ');

    ParseRule:=FALSE;

    exit;

   end;

   J:=Pos(SP[I],SVN);

   if J=0 then begin
    ShowMessage('Посторонний символ в правиле');

    ParseRule:=FALSE;

    exit;

   end;

   FVN:=TRUE;

   P[IdxVN,IdxVT]:=AddVN(P[IdxVN,IdxVT],VN[J]);

  end

  else begin

   if FVN then begin

    ShowMessage('Нарушена праволинейность грамматики, терминальный символ после нетерминального');

    ParseRule:=FALSE;

    exit;

   end;

   if IdxVT>0 then begin

    Ch:=GetNewVNSymbol;

    if Ch=#0 then begin

     ParseRule:=FALSE;

     exit;

    end;

    SVN:=SVN+Ch;

    Inc(NVN);

    VN[NVN]:=Ch;

    P[IdxVN,IdxVT]:=P[IdxVN,IdxVT]+Ch;

    IdxVN:=NVN;

   end;

   IdxVT:=J;

  end;

  Inc(I);

 end;

 if Not FVN then

  if IdxVT=0 then Final[IdxVN]:=TRUE

  else begin

   if IdxPFinal<0 then begin

    Ch:=GetNewVNSymbol;

    if Ch=#0 then begin

     ParseRule:=FALSE;

     exit;

    end;

    SVN:=SVN+Ch;

    Inc(NVN);

    IdxPFinal:=NVN;

    Final[NVN]:=TRUE;

    VN[NVN]:=Ch;

   end;

   P[IdxVN,IdxVT]:=AddVN(P[IdxVN,IdxVT],VN[IdxPFinal]);

  end;

 ParseRule:=TRUE;

end;

procedure TForm1.ReadGrammar; //процедура чтения грамматики с формы

var I, J: Integer;

begin

 IdxPFinal:=-1;

 for I:=1 to 26 do begin

  Final[I]:=FALSE;

  VN[I]:='';

  for J:=1 to 26 do P[I,J]:='';

 end;

 SVN:=Edit1.Text;

 NVN:=Length(SVN);

 SVT:=Edit2.Text;

 S0:=Edit3.Text;

 SEmpty:=Edit9.Text;

 SP:=Edit8.Text;

 if SEmpty='' then begin

  ShowMessage('Не задан пустой символ');

  exit;

 end;

 if Pos(SEmpty,SVN)<>0 then begin

  ShowMessage('Пустой символ встречен в нетерминальном алфавите');

  exit;

 end;

 if Pos(SEmpty,SVT)<>0 then begin

  ShowMessage('Пустой символ встречен в терминальном алфавите');

  exit;

 end;

 if SVN='' then begin

  ShowMessage('Не задан нетерминальный алфавит грамматики');

  exit;

 end;

 if SVT='' then begin

  ShowMessage('Не задан терминальный алфавит грамматики');

  exit;

 end;

 if Pos(S0,SVN)=0 then begin

  ShowMessage('Неправильно задан начальный символ грамматики');

  exit;

 end;

 if SP='' then begin

  ShowMessage('Не заданы правила грамматики');

  exit;

 end;

 for I:=1 to Length(SVN)-1 do

  if Pos(SVN[I],Copy(SVN,I+1,Length(SVN)))<>0 then begin

  ShowMessage('Повторное указание нетерминального символа');

  exit;

 end;

 for I:=1 to Length(SVT)-1 do

  if Pos(SVT[I],Copy(SVT,I+1,Length(SVT)))<>0 then begin

  ShowMessage('Повторное указание терминального символа');

  exit;

 end;

 for I:=1 to Length(SVN) do

  if Pos(SVN[I],SVT)<>0 then begin

  ShowMessage('Нетерминальный символ присутствует в терминальном алфавите');

  exit;

 end;

 for I:=1 to NVN do VN[I]:=SVN[I];

 while Pos(SEmpty,SP)<>0 do Delete(SP,Pos(SEmpty,SP),1);

end;

//процедура переименования двух и более буквенных имен  состояний в однобуквенные

procedure TForm1.RenameVN;

var NT: String; I, J, K, L: Integer;

begin

 NT:='A';

 for I:=1 to NVN do

  if Length(VN[I])>1 then begin

   J:=1;

   while J<=NVN do

    if NT=VN[J] then begin

     J:=1;

     NT:=Chr(Ord(NT[1])+1);

    end

    else

     Inc(J);

   for K:=1 to 26 do

    for L:=1 to 26 do

     if P[K,L]=VN[I] then P[K,L]:=NT;

   VN[I]:=NT;

   NT:=Chr(Ord(NT[1])+1);

  end;

end;

function TForm1.RGToNFA: Boolean; //функция преобразования грамматики в НКА

var IdxLVN, I, L: Integer;

begin

 I:=1;

 L:=Length(SP);

 while (I<L) do begin

  IdxLVN:=Pos(SP[I],SVN);

  if IdxLVN=0 then begin

   ShowMessage('Нетерминальный символ <'+SP[I]+'> не найден в алфавите');

   RGToNFA:=FALSE;

   exit;

  end;

  if SP[I+1]<>'=' then begin

   ShowMessage('Пропущен символ = после нетерминального символа');

   RGToNFA:=FALSE;

   exit;

  end;

  Inc(I,2);

  while ((I<=L) And (SP[I]<>';')) do begin

   if Not ParseRule(IdxLVN,I,L) then begin

    RGToNFA:=FALSE;

    exit;

   end;

   if ((I<=L) And (SP[I]='|')) then Inc(I);

  end;

  if Not Final[IdxLVN] then CheckFinal(IdxLVN);

  Inc(I);

 end;

 RGToNFA:=TRUE;

end;

function TForm1.SortString(S: String): String; //процедура лексиграфической сортировки символов строки

var I, J: Integer; Ch: Char;

begin

 for I:=1 to Length(S)-1 do

  for J:=I+1 to Length(S) do

   if (((S[I]>S[J]) And (S[I]<>S0[1])) Or (S[J]=S0[1])) then begin

    Ch:=S[J];

    S[J]:=S[I];

    S[I]:=Ch;

   end;

 SortString:=S;

end;

procedure TForm1.OnN4(Sender: TObject); //меню "Построить НКА"

begin

 ReadGrammar;

 if Not RGToNFA then exit;

 OutAutomat;

 Button1.Enabled:=FALSE;

end;

procedure TForm1.OnN2(Sender: TObject);

begin
 ShowMessage('Построение ДКА, эквивалентного заданной регулярной праволинейной грамматике');

end;

procedure TForm1.OnN3(Sender: TObject);

begin

 ShowMessage('Автор программы Рязанов Григорий Александрович');

end;

procedure TForm1.OnN5(Sender: TObject); //меню "Построить ДКА"

begin

 ReadGrammar;

 if Not RGToNFA then exit;

 OutAutomat;

 if Not CheckDFA then begin

  NFAToDFA;

  while DeleteUnavailable do ;

  OutAutomat;

 end;

 RenameVN;

 OutAutomat;

 Button1.Enabled:=TRUE;

end;

procedure TForm1.Button1Click(Sender: TObject); //кнопка "Проверить цепочку"

var I, J: Integer; SW, VN:String;

begin

 ListBox1.Items.Clear;

 VN:=Edit4.Text;

 SW:=Edit5.Text;

 I:=Pos(Edit6.Text,VN);

 while (SW<>'') do begin

  ListBox1.Items.Add('('+VN[I]+','+SW+')');

  J:=Pos(SW[1],SVT);

  if J=0 then begin

   ShowMessage('В цепочке посторонний символ');

   exit;

  end;

  if P[I,J]='' then begin

   ShowMessage('Нет перехода из '+VN[I]+' по символу '+SW[1]+'');

   exit;

  end;

  I:=Pos(P[I,J],VN);

  Delete(SW,1,1);

 end;

 ListBox1.Items.Add('('+VN[I]+','+SEmpty+')');

 if Pos(VN[I],Edit7.Text)=0 then

  ShowMessage('Цепочка отвергнута, ДКА не в конечном состоянии')

 else
  ShowMessage('Цепочка успешно прочитана');

end;

end.
5. РЕЗУЛЬТАТЫ РАБОТЫ ПРОГРАММЫ
На рис. 1 приведен пример работы программы с построенным ДКА для грамматики, описанной в данном отчете в примере из раздела 2.

Рис.1

На рис. 2 приведен пример работы программы с построенным ДКА для грамматики, описанной в примере в задании на курсовой проект.

